Enumerate States

Last updated 12/12/24

Enumerated States

- Enumerated States
  - Similar to enumerated types in C
  - Define a new TYPE with a fixed set of recognized values

```
type STATE_TYPE is (A, B, C, D, E, F, G, H, I, J);
```

  - Each state can be identified by its position (integer)
    - A – 0
    - B – 1

      …
    - J – 9

    - STATE_TYPE'pos(D) → 3

Enumerated States

- Enumerated States
  - Create signals of the enumerated type

```
signal state:      STATE_TYPE;
signal state_next: STATE_TYPE;
```

  - Synthesis is done using the enumerated states with the corresponding encoding
    - Binary
    - One Hot
    - Gray
    - Johnson
    - Auto

Enumerated States

Mod 10 Counter using Enumerated States



Enumerated States

- Mod 10 Counter – Enumerated States

Enumerated state declaration

State initialization and update (register logic)

Next state logic

```
----------------------------------------
--
-- counter_mod10_fsmC.vhdl
--
-- created 3/30/18
-- tj
--
-- rev 0
----------------------------------------
--
-- mod 10 up-counter - state machine with case statement
--                     enumerated states
----------------------------------------
--
-- Inputs: i_rstb, i_clk
-- Outputs: o_cnt[3:0]
--
----------------------------------------
library ieee;
use ieee.std_logic_1164.all;

entity counter_mod10_fsmC is
   port (
         i_clk :  in std_logic;
         i_rstb : in std_logic;

         o_cnt :  out std_logic_vector(3 downto 0)
    );
end entity;
```

```
architecture behavioral of counter_mod10_fsmC is

   --
   -- internal signals
   --
   type STATE_TYPE is (A, B, C, D, E, F, G, H, I, J);
   signal state:       STATE_TYPE;
   signal state_next:  STATE_TYPE;

begin

   --
   -- next state logic
   --
   process(state)
   begin
      case state is
         when A => state_next <= B;
         when B => state_next <= C;
         when C => state_next <= D;
         when D => state_next <= E;
         when E => state_next <= F;
         when F => state_next <= G;
         when G => state_next <= H;
         when H => state_next <= I;
         when I => state_next <= J;
         when J => state_next <= A;
         when others => state_next <= A;
      end case;
   end process;
```

```
   --
   -- Register logic
   --
   process(i_clk, i_rstb)
   begin
      -- reset
      if (i_rstb = '0') then
         state <= A;
      -- rising i_clk edge
      elsif (rising_edge (i_clk)) then
         state <= state_next;
      end if;
   end process;

   --
   -- Output logic
   --
   process(state)
   begin
      case state is
         when A => o_cnt <= "0000";
         when B => o_cnt <= "0001";
         when C => o_cnt <= "0010";
         when D => o_cnt <= "0011";
         when E => o_cnt <= "0100";
         when F => o_cnt <= "0101";
         when G => o_cnt <= "0110";
         when H => o_cnt <= "0111";
         when I => o_cnt <= "1000";
         when J => o_cnt <= "1001";
         when others => o_cnt <= "0000";
      end case;
   end process;

end behavioral;
```

Enumerated States

- Mod 10 Counter –Enumerated States



Quartus recognized part of the code as a state machine

Enumerated States

- Mod 10 Counter –Enumerated States





Enumerated States

- Mod 10 Counter –Enumerated States



	Source State	Destination State	Condition
1	A	B	
2	B	C	
3	C	D	
4	D	E	
5	E	F	
6	F	G	
7	G	H	
8	H	I	
9	I	J	
10	J	A	

Transitions / Encoding /

Enumerated States

- Mod 10 Counter –Enumerated States



Default encoding is a modified 1-hot

Transitions | Encoding

Enumerated States

- Mod 10 Counter –Enumerated States
  - Add the states and re-run the simulation